Pensieve header: Sep 22: Too many ways to compute Fibonacci.

ImageRestyle[ , ]



Too many ways to compute Fibonacci

The Naive Way

```
f0[0] = f0[1] = 1; f0[n_] := f0[n - 1] + f0[n - 2];
```

```
f0[10]
```

89

? Timing

Timing[*expr*] evaluates *expr*, and returns a list
of the time in seconds used, together with the result obtained. ≫

```
Timing[Log[Log[10 000 000!]] // N]
```

```
{3.03125, 18.834}
```

```
Table[i^2, {i, 0, 9}]
```

```
{0, 1, 4, 9, 16, 25, 36, 49, 64, 81}
```

```
Table[i^2, {i, 3, 9, 2}]
```

```
{9, 25, 49, 81}
```

? Table

Table[*expr*, *n*] generates a list of *n* copies of *expr*.
Table[*expr*, {*i*, i_{max}}] generates a list of the values of *expr* when *i* runs from 1 to i_{max}.
Table[*expr*, {*i*, i_{min}, i_{max}}] starts with $i = i_{min}$.
Table[*expr*, {*i*, i_{min}, i_{max}, *di*}] uses steps *di*.
Table[*expr*, {*i*, {i_1, i_2, ...}}] uses the successive values i_1, i_2, ...
Table[*expr*, {*i*, i_{min}, i_{max}}, {*j*, j_{min}, j_{max}}, ...]
gives a nested list. The list associated with *i* is outermost. ≫

```
Table[i j, {i, 10}, {j, 10}]
```

```
{{1, 2, 3, 4, 5, 6, 7, 8, 9, 10}, {2, 4, 6, 8, 10, 12, 14, 16, 18, 20},
 {3, 6, 9, 12, 15, 18, 21, 24, 27, 30}, {4, 8, 12, 16, 20, 24, 28, 32, 36, 40},
 {5, 10, 15, 20, 25, 30, 35, 40, 45, 50}, {6, 12, 18, 24, 30, 36, 42, 48, 54, 60},
 {7, 14, 21, 28, 35, 42, 49, 56, 63, 70}, {8, 16, 24, 32, 40, 48, 56, 64, 72, 80},
 {9, 18, 27, 36, 45, 54, 63, 72, 81, 90}, {10, 20, 30, 40, 50, 60, 70, 80, 90, 100}}
```

```
Table[i j, {i, 10}, {j, 10}] // MatrixForm
```

$$\begin{pmatrix} 1 & 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 \\ 2 & 4 & 6 & 8 & 10 & 12 & 14 & 16 & 18 & 20 \\ 3 & 6 & 9 & 12 & 15 & 18 & 21 & 24 & 27 & 30 \\ 4 & 8 & 12 & 16 & 20 & 24 & 28 & 32 & 36 & 40 \\ 5 & 10 & 15 & 20 & 25 & 30 & 35 & 40 & 45 & 50 \\ 6 & 12 & 18 & 24 & 30 & 36 & 42 & 48 & 54 & 60 \\ 7 & 14 & 21 & 28 & 35 & 42 & 49 & 56 & 63 & 70 \\ 8 & 16 & 24 & 32 & 40 & 48 & 56 & 64 & 72 & 80 \\ 9 & 18 & 27 & 36 & 45 & 54 & 63 & 72 & 81 & 90 \\ 10 & 20 & 30 & 40 & 50 & 60 & 70 & 80 & 90 & 100 \end{pmatrix}$$

? Echo

Echo[*expr*] prints *expr* and returns *expr*.
Echo[*expr*, *label*] prints *expr* prepending *label* and returns *expr*.
Echo[*expr*, *label*, *f*] prints *f*[*expr*] prepending *label* and returns *expr*. ≫

```
Echo[1 + 1]
```

```
2
```

```
2
```

```
Echo[1 + 1] * 7
```

```
2
```

```
14
```

```
Table[Echo@Timing[n → f0[n]], {n, 1, 40}]
```

```
{0., 1 → 1}
```

```
{0., 2 → 2}
```

```
{0., 3 → 3}
```

```
{0., 4 → 5}
```

```
{0., 5 → 8}
```

```
{0., 6 → 13}
```

```
{0., 7 → 21}
```

```
{0., 8 → 34}
```

```
{0., 9 → 55}
```

```
{0., 10 → 89}
```

```
{0., 11 → 144}
```

```
{0., 12 → 233}
```

```
{0., 13 → 377}
```

```
{0., 14 → 610}
```

```
{0.015625, 15 → 987}
```

```
{0., 16 → 1597}
```

```
{0., 17 → 2584}
```

```
{0., 18 → 4181}
```

```
{0., 19 → 6765}
```

```
{0.015625, 20 → 10 946}
```

```
{0.015625, 21 → 17 711}
```

```
{0.03125, 22 → 28 657}
```

```
{0.046875, 23 → 46 368}
```

```
{0.09375, 24 → 75 025}
```

```
{0.15625, 25 → 121 393}
```

{0.234375, 26 → 196 418}

{0.390625, 27 → 317 811}

{0.671875, 28 → 514 229}

{1.03125, 29 → 832 040}

{1.67188, 30 → 1 346 269}

{2.6875, 31 → 2 178 309}

{4.34375, 32 → 3 524 578}

{7.03125, 33 → 5 702 887}

{11.3594, 34 → 9 227 465}

{18.3438, 35 → 14 930 352}

{30.1094, 36 → 24 157 817}

{48.5469, 37 → 39 088 169}

{78.5313, 38 → 63 245 986}

$Aborted

The Naive Way, Corrected

```
sq[x_] := x^2
```

```
sq[7]
```

49

```
sq[7] = 50
```

50

```
? sq
```

Global`sq

sq[7] = 50

sq[x_] := x^2

```
Table[sq[i], {i, 10}]
```

{1, 4, 9, 16, 25, 36, 50, 64, 81, 100}

```
sq[x_] := (sq[x] = x^2)
```

```
? sq
```

```
Global`sq
```

```
sq[7] = 50
```

```
sq[x_] := sq[x] = x^2
```

```
Table[sq[x], {x, 10}]
```

```
{1, 4, 9, 16, 25, 36, 50, 64, 81, 100}
```

```
? sq
```

```
Global`sq
```

```
sq[1] = 1
```

```
sq[2] = 4
```

```
sq[3] = 9
```

```
sq[4] = 16
```

```
sq[5] = 25
```

```
sq[6] = 36
```

```
sq[7] = 50
```

```
sq[8] = 64
```

```
sq[9] = 81
```

```
sq[10] = 100
```

```
sq[x_] := sq[x] = x^2
```

```
f1[0] = f1[1] = 1; f1[n_] := (f1[n] = f1[n - 1] + f1[n - 2]);
```

```
f1[10]
```

```
89
```

```
? f1
```

```
Global`f1
```

```
f1[0] = 1
```

```
f1[1] = 1
```

```
f1[2] = 2
```

```
f1[3] = 3
```

```
f1[4] = 5
```

```
f1[5] = 8
```

```
f1[6] = 13
```

```
f1[7] = 21
```

```
f1[8] = 34
```

```
f1[9] = 55
```

```
f1[10] = 89
```

```
f1[n_] := f1[n] = f1[n - 1] + f1[n - 2]
```

```
Timing[f1[40]]
```

```
{0., 165 580 141}
```

```
Timing[f1[100]]
```

```
{0., 573 147 844 013 817 084 101}
```

```
f1[5]
```

```
8
```

```
f1[5] = 10
```

```
10
```

```
f1[5]
```

```
10
```

```
Table[f1[k], {k, 0, 10}]
```

```
{1, 1, 2, 3, 5, 10, 13, 21, 34, 55, 89}
```

```
f1[5] =.
```

```
f1[5]
```

```
8
```

"prev", "cur", and "While".

```
prev = 8; cur = 13
```

```
13
```

```
{prev = cur, cur = prev + cur}
```

```
{13, 26}
```

```
{prev = 8, cur = 13}
```

```
{8, 13}
```

```
{cur = prev + cur, prev = cur}
```

```
{21, 21}
```

```
{prev = 8, cur = 13}
```

```
{8, 13}
```

```
{prev, cur} = {cur, prev + cur}
```

```
{13, 21}
```

```
{a = 77, b = 55}
```

```
{77, 55}
```

```
a = b - a; b = b - a; a = a + b; {a, b}
```

```
{55, 77}
```

```
? While
```

```
While[test, body] evaluates test, then body, repetitively, until test first fails to give True.  >>
```

```
100!
```

```
93 326 215 443 944 152 681 699 238 856 266 700 490 715 968 264 381 621 468 592 963 895 217 599 993 229 915 ↘
 608 941 463 976 156 518 286 253 697 920 827 223 758 251 185 210 916 864 000 000 000 000 000 000 000 000
```

```
k = 1; p = 1;
While[k ≤ 100, p = p * k; k = k + 1];
p
```

```
93 326 215 443 944 152 681 699 238 856 266 700 490 715 968 264 381 621 468 592 963 895 217 599 993 229 915 ↘
 608 941 463 976 156 518 286 253 697 920 827 223 758 251 185 210 916 864 000 000 000 000 000 000 000 000
```

```
Range[100]
```

```
{1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21,
 22, 23, 24, 25, 26, 27, 28, 29, 30, 31, 32, 33, 34, 35, 36, 37, 38, 39, 40, 41,
 42, 43, 44, 45, 46, 47, 48, 49, 50, 51, 52, 53, 54, 55, 56, 57, 58, 59, 60, 61,
 62, 63, 64, 65, 66, 67, 68, 69, 70, 71, 72, 73, 74, 75, 76, 77, 78, 79, 80, 81,
 82, 83, 84, 85, 86, 87, 88, 89, 90, 91, 92, 93, 94, 95, 96, 97, 98, 99, 100}
```

```
Times @@ Range[100]
```

93 326 215 443 944 152 681 699 238 856 266 700 490 715 968 264 381 621 468 592 963 895 217 599 993 229 915 ⋱
608 941 463 976 156 518 286 253 697 920 827 223 758 251 185 210 916 864 000 000 000 000 000 000 000 000

```
k = 1; p = 1;
While[k ≤ 100, p *= k; k += 1];
p
```

93 326 215 443 944 152 681 699 238 856 266 700 490 715 968 264 381 621 468 592 963 895 217 599 993 229 915 ⋱
608 941 463 976 156 518 286 253 697 920 827 223 758 251 185 210 916 864 000 000 000 000 000 000 000 000

```
k = 1; p = 1;
While[k ≤ 100, p *= k; ++k];
p
```

93 326 215 443 944 152 681 699 238 856 266 700 490 715 968 264 381 621 468 592 963 895 217 599 993 229 915 ⋱
608 941 463 976 156 518 286 253 697 920 827 223 758 251 185 210 916 864 000 000 000 000 000 000 000 000

```
k = prev = cur = 1;
While[k < 100, {prev, cur} = {cur, prev + cur}; ++k];
cur
```

573 147 844 013 817 084 101

```
f2[n_] = (
  k = prev = cur = 1;
  While[k < n, {prev, cur} = {cur, prev + cur}; ++k];
  cur
 )
```

1

```
? ;
```

expr$_1$; *expr*$_2$; … evaluates the *expr*$_i$ in turn, giving the last one as the result. ≫

```
? =
```

lhs = *rhs* evaluates *rhs* and assigns the result to be the
value of *lhs*. From then on, *lhs* is replaced by *rhs* whenever it appears.
{*l*$_1$, *l*$_2$, …} = {*r*$_1$, *r*$_2$, …} evaluates the *r*$_i$, and assigns the results
to be the values of the corresponding *l*$_i$. ≫

```
Clear[f2]
```

```
f2[n_] = (
  {k, prev, cur} = {1, 1, 1};
  While[k < n, {prev, cur} = {cur, prev + cur}; ++k];
  cur
 )
```

1

```
f2[10]
```

1

```
Clear[f2]
```

```
f2[n_] := (
  {k, prev, cur} = {1, 1, 1};
  While[k < n, {prev, cur} = {cur, prev + cur}; ++k];
  cur
 )
```

```
f2[100]
```

573 147 844 013 817 084 101

```
prev
```

354 224 848 179 261 915 075

```
? Module
```

Module[{x, y, ...}, expr] specifies that
occurrences of the symbols x, y, ... in expr should be treated as local.
Module[{x = x0, ...}, expr] defines initial values for x, ... >>

```
f3[n_] := Module[{k, prev, cur},
  {k, prev, cur} = {1, 1, 1};
  While[k < n, {prev, cur} = {cur, prev + cur}; ++k];
  cur
 ]
```

```
f3[40]
```

165 580 141

```
cur
```

573 147 844 013 817 084 101

Aside on @, @@, @@@

```
Times @ Range[5]
```

{1, 2, 3, 4, 5}

```
Times @@ Range[5]
```

120

```
Times[2, 3, 5]
```

30

```
Times[2, 3]
```

```
6
```

```
Times[2]
```

```
2
```

```
Times[]
```

```
1
```

```
Plus[]
```

```
0
```

```
h @@@ f[g[1], g[2]]
```

```
f[h[1], h[2]]
```

"prev", "cur", and "For".

```
? For
```

For[*start*, *test*, *incr*, *body*] executes *start*, then
repeatedly evaluates *body* and *incr* until *test* fails to give True. ≫

```
For[k = cur = prev = 1, k < 100, ++k, {prev, cur} = {cur, prev + cur}]; cur
```

```
573 147 844 013 817 084 101
```

```
3 == 7
```

```
False
```

```
7 == 7
```

```
True
```

"prev", "cur", and "Do".

```
? Do
```

Do[*expr*, *n*] evaluates *expr* *n* times.
Do[*expr*, {*i*, i_{max}}] evaluates *expr* with the variable
i successively taking on the values 1 through i_{max} (in steps of 1).
Do[*expr*, {*i*, i_{min}, i_{max}}] starts with $i = i_{min}$.
Do[*expr*, {*i*, i_{min}, i_{max}, *di*}] uses steps *di*.
Do[*expr*, {*i*, {i_1, i_2, ...}}] uses the successive values i_1, i_2, ...
Do[*expr*, {*i*, i_{min}, i_{max}}, {*j*, j_{min}, j_{max}}, ...]
evaluates *expr* looping over different values of *j* etc. for each *i*. ≫

```
{prev, cur} = {1, 1};
Do[{prev, cur} = {cur, prev + cur}, 99];
cur
```

573 147 844 013 817 084 101

A "While" loop for $\{f_1, f_2, \ldots\}$ (using negative indices)

```
Length[{1, 2, 4}]
```

3

```
? Append
```

Append[*expr*, *elem*] gives *expr* with *elem* appended.
Append[*elem*] represents an operator form of Append that can be applied to an expression. >>

```
fs = Append[f[1, 2, 3], "tristan"]
```

f[1, 2, 3, tristan]

```
fs[[4]]
```

tristan

```
fs[[2]]
```

2

```
fs[[-1]]
```

tristan

```
fs[[0]]
```

f

```
First[fs]
```

1

```
Last[fs]
```

tristan

```
Head[fs]
```

f

```
fs = {1, 1};
f6[n_] := Module[{fs = {1, 1}},
  While[Length[fs] ≤ n, Append[fs, fs[[-1]] + fs[[-2]]]];
  Last[fs]
 ]
```

```
f6[100]
```

```
$Aborted
```

```
fs = {1, 1}
```

```
{1, 1}
```

```
While[Length[fs] ≤ 5, fs = Append[fs, fs[[-1]] + fs[[-2]]]]
```

```
fs
```

```
{1, 1, 2, 3, 5, 8}
```

```
fs = {1, 1};
f6[n_] := Module[{fs = {1, 1}},
  While[Length[fs] ≤ n, fs = Append[fs, fs[[-1]] + fs[[-2]]]];
  Last[fs]
 ]
```

```
f6[100]
```

```
573 147 844 013 817 084 101
```

```
fs = {1, 1};
f7[n_] := Module[{fs = {1, 1}},
  While[Length[fs] ≤ n, fs = Append[fs, fs[[-1]] + fs[[-2]]]];
  fs
 ]
```

```
f7[20]
```

```
{1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765, 10 946}
```

```
fs = {1, 1};
f8[n_] := Module[{fs = {1, 1}},
  While[Length[fs] ≤ n, AppendTo[fs, fs[[-1]] + fs[[-2]]]];
  fs
 ]
```

```
f8[10]
```

```
{1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89}
```

A "While" loop for $\{f_1, f_2, \ldots\}$ (using "Total" and "Most")

```
fs = {1, 1};
f9[n_] := Module[{fs = {1, 1}},
  While[Length[fs] ≤ n, AppendTo[fs, 1 + Total@Drop[fs, -1]]];
  fs
 ]
```

```
f9[10]
```

```
{1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89}
```

```
fs = {1, 1};
f10[n_] := Module[{fs = {1, 1}},
  While[Length[fs] ≤ n, AppendTo[fs, 1 + Total@Most@fs]];
  fs
 ]
```

```
f10[10]
```

```
{1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89}
```

```
Total[Range[100]]
```

```
5050
```

```
Plus @@ Range[100]
```

```
5050
```

"ReplaceRepeated" on $\begin{pmatrix} n \\ f_{n-1} \\ f_n \end{pmatrix}$.

"NestWhile" on $\begin{pmatrix} n \\ f_{n-1} \\ f_n \end{pmatrix}$.

"Nest" on $\begin{pmatrix} f_{n-1} \\ f_n \end{pmatrix}$.

A Sum of Binomial Coefficients

Solve for an "explicit" formula, then use it.

"Series" and $\frac{1}{1-x-x^2}$

```
Series[1/(1 - x - x^2), {x, 0, 10}]
```

$$1 + x + 2\,x^2 + 3\,x^3 + 5\,x^4 + 8\,x^5 + 13\,x^6 + 21\,x^7 + 34\,x^8 + 55\,x^9 + 89\,x^{10} + O[x]^{11}$$

Other Items

Continue looking at Charlene's project? A look at Etienne's project?